
Komega Documentation

リリース *0.1*

kawamura

2017 年 01 月 10 日

Contents

1	概要	2
2	著作権	3
3	アルゴリズム	4
3.1	Seed switch 付き Shifted BiCG 法	4
3.2	Seed switch 付き Shifted COCG 法	5
3.3	Seed switch 付き Shifted CG 法	6
4	プログラム内でのライブラリの動作イメージ	8
4.1	Shifted BiCG ライブラリの動作イメージ	9
4.2	Shifted COCG ライブラリの動作イメージ	10
4.3	Shifted CG ライブラリの動作イメージ	11
5	使用方法	12
5.1	各ルーチンの説明	13
5.2	Shifted BiCG ライブラリを使用したソースコードの例	20
6	Contact	24
	Bibliography	25

Contents:

Chapter 1

概要

本資料は ISSP Math Library の内の、Krylov 部分空間法に基づくシフト線形方程式群ソルバーライブラリ $K(\omega)$ に関するマニュアルである。本ライブラリは、(射影付き) シフト線形問題

$$G_{ij}(z) = \langle i | (z\hat{I} - \hat{H})^{-1} | j \rangle \equiv \varphi_i^* \cdot (z\hat{I} - \hat{H})^{-1} \varphi_j \quad (1.1)$$

を、Krylov 部分空間法を用いて解くためのルーチンを提供する。言語は fortran を用いる。また、BLAS レベル 1 ルーチンを使用する。

Chapter 2

著作権

This software is developed under the support of
“*Project for advancement of software usability in materials science*” by The
Institute for Solid State Physics, The University of Tokyo.

This library is free software; you can redistribute it and/or
modify it under the terms of the GNU Lesser General Public
License as published by the Free Software Foundation; either
version 2.1 of the License, or (at your option) any later version.

This library is distributed in the hope that it will be useful,
but WITHOUT ANY WARRANTY; without even the implied warranty of
MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE. See the GNU
Lesser General Public License for more details.

You should have received a copy of the GNU Lesser General Public
License along with this library; if not, write to the Free Software
Foundation, Inc., 59 Temple Place, Suite 330, Boston, MA 02111-1307 USA

For more details, See ‘COPYING.LESSER’ in the root directory of this library.

Chapter 3

アルゴリズム

このライブラリは, \hat{H} および z が複素数であるか実数であるかに応じて, 次の 4 種類の計算をサポートする (\hat{H} は複素数の場合はエルミート行列, 実数の場合は実対称行列).

- \hat{H} も z も両方複素数の場合 : Shifted Bi-Conjugate Gradient(BiCG) 法 [Flommer2003]
- \hat{H} が実数で z が複素数の場合 : Shifted Conjugate Orthogonal Conjugate Gradient(COCG) 法 [Yamamoto2008]
- \hat{H} が複素数で z が実数の場合 : Shifted Conjugate Gradient(CG) 法 (複素ベクトル)
- \hat{H} も z も両方実数の場合 : Shifted Conjugate Gradient(CG) 法 (実ベクトル)

いずれの場合も Seed switching [Yamamoto2008] を行う. 左ベクトルが N_L 個, 右ベクトルが N_R 個 (典型的には 1 個) あるとする. 以下, 各手法のアルゴリズムを記載する.

3.1 Seed switch 付き Shifted BiCG 法

$$G_{ij}(z_k) = 0 (i = 1 \cdots N_L, j = 1 \cdots N_R, k = 1 \cdots N_z)$$

do $j = 1 \cdots N_R$

$$\mathbf{r} = \boldsymbol{\varphi}_j,$$

$$\tilde{\mathbf{r}} = \text{任意}, \mathbf{r}^{\text{old}} = \tilde{\mathbf{r}}^{\text{old}} = \mathbf{0}$$

$$p_{ik} = 0 (i = 1 \cdots N_L, k = 1 \cdots N_z), \pi_k = \pi_k^{\text{old}} = 1 (k = 1 \cdots N_z)$$

$$\rho = \infty, \alpha = 1, z_{\text{seed}} = 0$$

do iteration

◦ シード方程式

$$\rho^{\text{old}} = \rho, \rho = \tilde{\mathbf{r}}^* \cdot \mathbf{r}$$

$$\beta = \rho / \rho^{\text{old}}$$

$$\mathbf{q} = (z_{\text{seed}} \hat{I} - \hat{H}) \mathbf{r}$$

$$\alpha^{\text{old}} = \alpha, \alpha = \frac{\rho}{\tilde{\mathbf{r}}^* \cdot \mathbf{q} - \beta \rho / \alpha}$$

◦ シフト方程式

do $k = 1 \cdots N_z$

```

 $\pi_k^{\text{new}} = [1 + \alpha(z_k - z_{\text{seed}})]\pi_k - \frac{\alpha\beta}{\alpha^{\text{old}}}(\pi_k^{\text{old}} - \pi_k)$ 
do  $i = 1 \cdots N_L$ 
   $p_{ik} = \frac{1}{\pi_k} \boldsymbol{\varphi}_i^* \cdot \mathbf{r} + \frac{\pi_k^{\text{old}} \pi_k^{\text{old}}}{\pi_k \pi_k} \beta p_{ik}$ 
   $G_{ij}(z_k) = G_{ij}(z_k) + \frac{\pi_k}{\pi_k^{\text{new}}} \alpha p_{ik}$ 
   $\pi_k^{\text{old}} = \pi_k, \pi_k = \pi_k^{\text{new}}$ 
end do  $i$ 
end do  $k$ 
 $\mathbf{q} = \left(1 + \frac{\alpha\beta}{\alpha^{\text{old}}}\right) \mathbf{r} - \alpha \mathbf{q} - \frac{\alpha\beta}{\alpha^{\text{old}}} \mathbf{r}^{\text{old}}, \mathbf{r}^{\text{old}} = \mathbf{r}, \mathbf{r} = \mathbf{q}$ 
 $\mathbf{q} = (z_{\text{seed}}^* \hat{I} - \hat{H}) \tilde{\mathbf{r}}, \mathbf{q} = \left(1 + \frac{\alpha^* \beta^*}{\alpha^{\text{old}*}}\right) \tilde{\mathbf{r}} - \alpha^* \mathbf{q} - \frac{\alpha^* \beta^*}{\alpha^{\text{old}*}} \tilde{\mathbf{r}}^{\text{old}}, \tilde{\mathbf{r}}^{\text{old}} = \tilde{\mathbf{r}}, \tilde{\mathbf{r}} = \mathbf{q}$ 
○ Seed switch
 $|\pi_k|$  が最も小さい  $k$  を探す.  $\rightarrow z_{\text{seed}}, \pi_{\text{seed}}, \pi_{\text{seed}}^{\text{old}}$ 
 $\mathbf{r} = \mathbf{r}/\pi_{\text{seed}}, \mathbf{r}^{\text{old}} = \mathbf{r}^{\text{old}}/\pi_{\text{seed}}^{\text{old}}, \tilde{\mathbf{r}} = \tilde{\mathbf{r}}/\pi_{\text{seed}}^*, \tilde{\mathbf{r}}^{\text{old}} = \tilde{\mathbf{r}}^{\text{old}}/\pi_{\text{seed}}^{\text{old}*}$ 
 $\alpha = (\pi_{\text{seed}}^{\text{old}}/\pi_{\text{seed}})\alpha, \rho = \rho/(\pi_{\text{seed}}^{\text{old}} \pi_{\text{seed}}^{\text{old}})$ 
 $\{\pi_k = \pi_k/\pi_{\text{seed}}, \pi_k^{\text{old}} = \pi_k^{\text{old}}/\pi_{\text{seed}}^{\text{old}}\}$ 
if(  $|\mathbf{r}| < \text{Threshold}$ ) exit
end do iteration
end do  $j$ 

```

3.2 Seed switch 付き Shifted COCG 法

BiCG のアルゴリズムで, $\tilde{\mathbf{r}} = \mathbf{r}^*, \tilde{\mathbf{r}}^{\text{old}} = \mathbf{r}^{\text{old}*}$ とすると得られる.

$G_{ij}(z_k) = 0 (i = 1 \cdots N_L, j = 1 \cdots N_R, k = 1 \cdots N_z)$

do $j = 1 \cdots N_R$

$\mathbf{r} = \boldsymbol{\varphi}_j, \mathbf{r}^{\text{old}} = \mathbf{0}$

$p_{ik} = 0 (i = 1 \cdots N_L, k = 1 \cdots N_z), \pi_k = \pi_k^{\text{old}} = 1 (k = 1 \cdots N_z)$

$\rho = \infty, \alpha = 1, z_{\text{seed}} = 0$

do iteration

○ シード方程式

$\rho^{\text{old}} = \rho, \rho = \mathbf{r} \cdot \mathbf{r}$

$\beta = \rho/\rho^{\text{old}}$

$\mathbf{q} = (z_{\text{seed}} \hat{I} - \hat{H}) \mathbf{r}$

$\alpha^{\text{old}} = \alpha, \alpha = \frac{\rho}{\mathbf{r} \cdot \mathbf{q} - \beta \rho / \alpha}$

○ シフト方程式

do $k = 1 \cdots N_z$

```


$$\pi_k^{\text{new}} = [1 + \alpha(z_k - z_{\text{seed}})]\pi_k - \frac{\alpha\beta}{\alpha^{\text{old}}}(\pi_k^{\text{old}} - \pi_k)$$

do  $i = 1 \cdots N_L$ 
    
$$p_{ik} = \frac{1}{\pi_k} \boldsymbol{\varphi}_i^* \cdot \mathbf{r} + \frac{\pi_k^{\text{old}} \pi_k^{\text{old}}}{\pi_k \pi_k} \beta p_{ik}$$

    
$$G_{ij}(z_k) = G_{ij}(z_k) + \frac{\pi_k}{\pi_k^{\text{new}}} \alpha p_{ik}$$

    
$$\pi_k^{\text{old}} = \pi_k, \pi_k = \pi_k^{\text{new}}$$

end do  $i$ 
end do  $k$ 

$$\mathbf{q} = \left(1 + \frac{\alpha\beta}{\alpha^{\text{old}}}\right) \mathbf{r} - \alpha \mathbf{q} - \frac{\alpha\beta}{\alpha^{\text{old}}} \mathbf{r}^{\text{old}}, \mathbf{r}^{\text{old}} = \mathbf{r}, \mathbf{r} = \mathbf{q}$$

◦ Seed switch
 $|\pi_k|$  が最も小さい  $k$  を探す.  $\rightarrow z_{\text{seed}}, \pi_{\text{seed}}, \pi_{\text{seed}}^{\text{old}}$ 

$$\mathbf{r} = \mathbf{r} / \pi_{\text{seed}}, \mathbf{r}^{\text{old}} = \mathbf{r}^{\text{old}} / \pi_{\text{seed}}^{\text{old}}$$


$$\alpha = (\pi_{\text{seed}}^{\text{old}} / \pi_{\text{seed}}) \alpha, \rho = \rho / (\pi_{\text{seed}}^{\text{old}} \pi_{\text{seed}}^{\text{old}})$$


$$\{\pi_k = \pi_k / \pi_{\text{seed}}, \pi_k^{\text{old}} = \pi_k^{\text{old}} / \pi_{\text{seed}}^{\text{old}}\}$$

if(  $|\mathbf{r}| < \text{Threshold}$ ) exit
end do iteration
end do  $j$ 

```

3.3 Seed switch 付き Shifted CG 法

BiCG のアルゴリズムで, $\tilde{\mathbf{r}} = \mathbf{r}, \tilde{\mathbf{r}}^{\text{old}} = \mathbf{r}^{\text{old}}$ とすると得られる.

$G_{ij}(z_k) = 0 (i = 1 \cdots N_L, j = 1 \cdots N_R, k = 1 \cdots N_z)$

do $j = 1 \cdots N_R$

$\mathbf{r} = \boldsymbol{\varphi}_j, \mathbf{r}^{\text{old}} = \mathbf{0}$

$p_{ik} = 0 (i = 1 \cdots N_L, k = 1 \cdots N_z), \pi_k = \pi_k^{\text{old}} = 1 (k = 1 \cdots N_z)$

$\rho = \infty, \alpha = 1, z_{\text{seed}} = 0$

do iteration

◦ シード方程式

$\rho^{\text{old}} = \rho, \rho = \mathbf{r}^* \cdot \mathbf{r}$

$\beta = \rho / \rho^{\text{old}}$

$\mathbf{q} = (z_{\text{seed}} \hat{I} - \hat{H}) \mathbf{r}$

$\alpha^{\text{old}} = \alpha, \alpha = \frac{\rho}{\mathbf{r}^* \cdot \mathbf{q} - \beta \rho / \alpha}$

◦ シフト方程式

do $k = 1 \cdots N_z$

$$\pi_k^{\text{new}} = [1 + \alpha(z_k - z_{\text{seed}})]\pi_k - \frac{\alpha\beta}{\alpha^{\text{old}}}(\pi_k^{\text{old}} - \pi_k)$$

do $i = 1 \cdots N_L$

$$p_{ik} = \frac{1}{\pi_k} \boldsymbol{\varphi}_i^* \cdot \mathbf{r} + \left(\frac{\pi_k^{\text{old}}}{\pi_k} \right)^2 \beta p_{ik}$$

$$G_{ij}(z_k) = G_{ij}(z_k) + \frac{\pi_k}{\pi_k^{\text{new}}} \alpha p_{ik}$$

$$\pi_k^{\text{old}} = \pi_k, \pi_k = \pi_k^{\text{new}}$$

end do i

end do k

$$\mathbf{q} = \left(1 + \frac{\alpha\beta}{\alpha^{\text{old}}} \right) \mathbf{r} - \alpha \mathbf{q} - \frac{\alpha\beta}{\alpha^{\text{old}}} \mathbf{r}^{\text{old}}, \mathbf{r}^{\text{old}} = \mathbf{r}, \mathbf{r} = \mathbf{q}$$

◦ Seed switch

$|\pi_k|$ が最も小さい k を探す. $\rightarrow z_{\text{seed}}, \pi_{\text{seed}}, \pi_{\text{seed}}^{\text{old}}$

$$\mathbf{r} = \mathbf{r} / \pi_{\text{seed}}, \mathbf{r}^{\text{old}} = \mathbf{r}^{\text{old}} / \pi_{\text{seed}}^{\text{old}}$$

$$\alpha = (\pi_{\text{seed}}^{\text{old}} / \pi_{\text{seed}}) \alpha, \rho = \rho / \pi_{\text{seed}}^{\text{old}^2}$$

$$\{\pi_k = \pi_k / \pi_{\text{seed}}, \pi_k^{\text{old}} = \pi_k^{\text{old}} / \pi_{\text{seed}}^{\text{old}}\}$$

if($|\mathbf{r}| < \text{Threshold}$) exit

end do iteration

end do j

Chapter 4

プログラム内でのライブラリの動作イメージ

以下では N_R のループは省略する (各右辺ベクトルごとに同じ事をすればいいので). また $G_{ij}(z_k)$ の代わりに N_z 個の N_L 次元の解ベクトル \mathbf{x}_k を求める.

注意事項.

- `komega_????_init` を呼び出す前にサイズ N_H のベクトルを 2 本 (BiCG の時には 4 本) Allocate しておく.
- ハミルトニアン-ベクトル積を行う部分はあらかじめ作成しておく.
- 解ベクトルを Allocate しておく. ただし, これは必ずしも $N_H \times N_z$ である必要はない. 実際前節の場合は $N_L \times N_z$ である. この時 (双) 共役勾配ベクトル \mathbf{p}_k も N_z 個の N_L 次元のベクトルである. N_H 次元の残差ベクトルを N_L 次元へ変換するところはユーザーが行う.

$$\mathbf{r}^L = \hat{P}^\dagger \mathbf{r}, \quad \hat{P} \equiv (\varphi_1, \dots, \varphi_{N_L})$$

- `komega_????_update` の出力 `status` の第一成分が負の値になった場合には, 解が収束した, 若しくは破たんしたことを表す. したがって `status(1) < 0` でループを抜けるようにしておく.
- `komega_????_update` 内での収束判定には, シード点での残差ベクトルの 2-ノルムが使われる. すなわち, すべてのシフト点での残差ベクトルの 2-ノルムが `threshold` を下回った時に収束したと見做される.
- 各 iteration での $\alpha, \beta, \mathbf{r}^L$ を保存しておき, あとで利用する場合には最大反復回数 “`itermax`” を設定する.

ライブラリの各ルーチンの名前は次の通りである.

- `komega_BiCG_init, komega_CO CG_init, komega_CG_C_init, komega_CG_R_init`
ライブラリ内部で使う (ユーザーの目に触れない) 変数の Allocate や初期値設定を行う.
- `komega_BiCG_update, komega_CO CG_update, komega_CG_C_update, komega_CG_R_update`
Iteration の中で呼び出される. 解ベクトル群の更新等を行う.
- `komega_BiCG_finalize, komega_CO CG_finalize, komega_CG_C_finalize, komega_CG_R_finalize`
Allocate したライブラリ内部ベクトルを開放する.

- komega_BiCG_getcoef, komega_CO CG_getcoef, komega_CG_C_getcoef,
komega_CG_R_getcoef

各 iteration で保存しておいた $\alpha, \beta, z_{\text{seed}}, \mathbf{r}^L$ を取り出す.

- komega_BiCG_getvec, komega_CO CG_getvec, komega_CG_C_getvec,
komega_CG_R_getvec

$\mathbf{r}, \mathbf{r}^{\text{old}}, \tilde{\mathbf{r}}, \tilde{\mathbf{r}}^{\text{old}}$ を取り出す.

- komega_BiCG_restart, komega_CO CG_restart, komega_CG_C_restart,
komega_CG_R_restart

保存しておいた α 等を用いて, 新規の z での計算を行う. \mathbf{r} 等も有る場合には komega_BiCG_init, komega_CO CG_init, komega_CG_C_init, komega_CG_R_init の代わりに用いてリスタートすることもできる.

4.1 Shifted BiCG ライブラリの動作イメージ

Allocate $\mathbf{v}_{12}, \mathbf{v}_{13}, \mathbf{v}_2, \mathbf{v}_3, \{\mathbf{x}_k\}, \mathbf{r}^L, \mathbf{v}_2 = \varphi_j$

komega_BiCG_init(N_H, N_L, N_z, x, z, itermx, threshold) start

Allocate $\mathbf{v}_3, \mathbf{v}_5, \{\pi_k\}, \{\pi_k^{\text{old}}\}, \{\mathbf{p}_k\}$

Copy $\{z_k\}$

itermx $\neq 0$ ならば $\alpha, \beta, \mathbf{r}^L$ を保存する配列を確保する.

$\mathbf{v}_4 = \mathbf{v}_2^*$ (任意), $\mathbf{v}_3 = \mathbf{v}_5 = \mathbf{0}$,

$\mathbf{p}_k = \mathbf{x}_k = \mathbf{0} (k = 1 \cdots N_z), \pi_k = \pi_k^{\text{old}} = 1 (k = 1 \cdots N_z)$

$\rho = \infty, \alpha = 1, z_{\text{seed}} = 0$

($\mathbf{v}_2 \equiv \mathbf{r}, \mathbf{v}_3 \equiv \mathbf{r}^{\text{old}}, \mathbf{v}_4 \equiv \tilde{\mathbf{r}}, \mathbf{v}_5 \equiv \tilde{\mathbf{r}}^{\text{old}}$.)

komega_BiCG_init finish

do iteration

$\mathbf{r}^L = \hat{P}^\dagger \mathbf{v}_2$

$\mathbf{v}_{12} = \hat{H} \mathbf{v}_2, \mathbf{v}_{14} = \hat{H} \mathbf{v}_4$ [($\mathbf{v}_{12}, \mathbf{v}_{14}$) = $\hat{H}(\mathbf{v}_2, \mathbf{v}_4)$ とも書ける]

komega_BiCG_update(v_12, v_2, v_14, v_4, x, r_small, status) start

◦ シード方程式

$\rho^{\text{old}} = \rho, \rho = \mathbf{v}_4^* \cdot \mathbf{v}_2$

$\beta = \rho / \rho^{\text{old}}$

$\mathbf{v}_{12} = z_{\text{seed}} \mathbf{v}_2 - \mathbf{v}_{12}, \mathbf{v}_{14} = z_{\text{seed}}^* \mathbf{v}_4 - \mathbf{v}_{14}$

$\alpha^{\text{old}} = \alpha, \alpha = \frac{\rho}{\mathbf{v}_3^* \cdot \mathbf{v}_{12} - \beta \rho / \alpha}$

◦ シフト方程式

do $k = 1 \cdots N_z$

$\pi_k^{\text{new}} = [1 + \alpha(z_k - z_{\text{seed}})]\pi_k - \frac{\alpha\beta}{\alpha^{\text{old}}}(\pi_k^{\text{old}} - \pi_k)$

$\mathbf{p}_k = \frac{1}{\pi_k} \mathbf{r}^L + \frac{\pi_k^{\text{old}} \pi_k^{\text{old}}}{\pi_k \pi_k} \beta \mathbf{p}_k$

```


$$\mathbf{x}_k = \mathbf{x}_k + \frac{\pi_k}{\pi_k^{\text{new}}} \alpha \mathbf{p}_k$$


$$\pi_k^{\text{old}} = \pi_k, \pi_k = \pi_k^{\text{new}}$$

end do  $k$ 

$$\mathbf{v}_{12} = \left(1 + \frac{\alpha\beta}{\alpha^{\text{old}}}\right) \mathbf{v}_2 - \alpha \mathbf{v}_{12} - \frac{\alpha\beta}{\alpha^{\text{old}}} \mathbf{v}_3, \mathbf{v}_3 = \mathbf{v}_2, \mathbf{v}_2 = \mathbf{v}_{12}$$


$$\mathbf{v}_{14} = \left(1 + \frac{\alpha^* \beta^*}{\alpha^{\text{old}*}}\right) \mathbf{v}_4 - \alpha^* \mathbf{v}_{14} - \frac{\alpha^* \beta^*}{\alpha^{\text{old}*}} \mathbf{v}_5, \mathbf{v}_5 = \mathbf{v}_4, \mathbf{v}_4 = \mathbf{v}_{14}$$

◦ Seed switch
 $|\pi_k|$  が最も小さい  $k$  を探す.  $\rightarrow z_{\text{seed}}, \pi_{\text{seed}}, \pi_{\text{seed}}^{\text{old}}$ 
 $\mathbf{v}_2 = \mathbf{v}_2 / \pi_{\text{seed}}, \mathbf{v}_3 = \mathbf{v}_3 / \pi_{\text{seed}}^{\text{old}}, \mathbf{v}_4 = \mathbf{v}_4 / \pi_{\text{seed}}^*, \mathbf{v}_5 = \mathbf{v}_5 / \pi_{\text{seed}}^{\text{old}*}$ 
 $\alpha = (\pi_{\text{seed}}^{\text{old}} / \pi_{\text{seed}}) \alpha, \rho = \rho / (\pi_{\text{seed}}^{\text{old}} \pi_{\text{seed}}^{\text{old}})$ 
 $\{\pi_k = \pi_k / \pi_{\text{seed}}, \pi_k^{\text{old}} = \pi_k^{\text{old}} / \pi_{\text{seed}}^{\text{old}}\}$ 
komega_BiCG_update finish

if(status(1) < 0 (これは:  $|\mathbf{v}_2| < \text{Threshold}$  となった事を意味する)) exit
end do iteration

komega_BiCG_finalize start
  Deallocate  $\mathbf{v}_4, \mathbf{v}_5, \{\pi_k\}, \{\pi_k^{\text{old}}\}, \{\mathbf{p}_k\}$ 
komega_BiCG_finalize finish

```

4.2 Shifted COCG ライブラリの動作イメージ

```

Allocate  $\mathbf{v}_1, \mathbf{v}_2, \{\mathbf{x}_k\}, \mathbf{r}^L, \mathbf{v}_2 = \boldsymbol{\varphi}_j$ 
COCG_init(N_H, N_L, N_z, x, z, itemax, threshold) start
  Allocate  $\mathbf{v}_3, \{\pi_k\}, \{\pi_k^{\text{old}}\}, \{\mathbf{p}_k\}$ 
  Copy  $\{z_k\}$ 
  itemax  $\neq 0$  ならば  $\alpha, \beta, \mathbf{r}^L$  を保存する配列を確保する.
   $\mathbf{v}_3 = \mathbf{0},$ 
   $\mathbf{p}_k = \mathbf{x}_k = \mathbf{0} (k = 1 \cdots N_z), \pi_k = \pi_k^{\text{old}} = 1 (k = 1 \cdots N_z)$ 
   $\rho = \infty, \alpha = 1, \beta = 0, z_{\text{seed}} = 0$ 
  ( $\mathbf{v}_2 \equiv \mathbf{r}, \mathbf{v}_3 \equiv \mathbf{r}^{\text{old}}.$ )
COCG_init finish

do iteration
   $\mathbf{r}^L = \hat{P}^\dagger \mathbf{v}_2$ 
   $\mathbf{v}_1 = \hat{H} \mathbf{v}_2$ 
  COCG_update(v_1, v_2, x, r_small, status) start
    ◦ シード方程式
     $\rho^{\text{old}} = \rho, \rho = \mathbf{v}_2 \cdot \mathbf{v}_2$ 
     $\beta = \rho / \rho^{\text{old}}$ 

```

$$\mathbf{v}_1 = z_{\text{seed}} \mathbf{v}_2 - \mathbf{v}_1$$

$$\alpha^{\text{old}} = \alpha, \quad \alpha = \frac{\rho}{\mathbf{v}_2 \cdot \mathbf{v}_1 - \beta \rho / \alpha}$$

◦ シフト方程式

do $k = 1 \cdots N_z$

$$\pi_k^{\text{new}} = [1 + \alpha(z_k - z_{\text{seed}})]\pi_k - \frac{\alpha\beta}{\alpha^{\text{old}}}(\pi_k^{\text{old}} - \pi_k)$$

$$\mathbf{p}_k = \frac{1}{\pi_k} \mathbf{r}^L + \frac{\pi_k^{\text{old}} \pi_k^{\text{old}}}{\pi_k \pi_k} \beta \mathbf{p}_k$$

$$\mathbf{x}_k = \mathbf{x}_k + \frac{\pi_k}{\pi_k^{\text{new}}} \alpha \mathbf{p}_k$$

$$\pi_k^{\text{old}} = \pi_k, \quad \pi_k = \pi_k^{\text{new}}$$

end do k

$$\mathbf{v}_1 = \left(1 + \frac{\alpha\beta}{\alpha^{\text{old}}}\right) \mathbf{v}_2 - \alpha \mathbf{v}_1 - \frac{\alpha\beta}{\alpha^{\text{old}}} \mathbf{v}_3$$

$$\mathbf{v}_3 = \mathbf{v}_2, \mathbf{v}_2 = \mathbf{v}_1$$

◦ Seed switch

$|\pi_k|$ が最も小さい k を探す. $\rightarrow z_{\text{seed}}, \pi_{\text{seed}}, \pi_{\text{seed}}^{\text{old}}$

$$\mathbf{v}_2 = \mathbf{v}_2 / \pi_{\text{seed}}, \mathbf{v}_3 = \mathbf{v}_3 / \pi_{\text{seed}}^{\text{old}}$$

$$\alpha = (\pi_{\text{seed}}^{\text{old}} / \pi_{\text{seed}}) \alpha, \quad \rho = \rho / (\pi_{\text{seed}}^{\text{old}} \pi_{\text{seed}}^{\text{old}})$$

$$\{\pi_k = \pi_k / \pi_{\text{seed}}, \quad \pi_k^{\text{old}} = \pi_k^{\text{old}} / \pi_{\text{seed}}^{\text{old}}\}$$

COCG_update finish

if(status(1) < 0 (これは: $\|\mathbf{v}\|_2 < \text{Threshold}$ となった事を意味する)) exit

end do iteration

COCG_finalize start

Deallocate $\mathbf{v}_3, \{\pi_k\}, \{\pi_k^{\text{old}}\}, \{\mathbf{p}_k\}$

COCG_finalize finish

4.3 Shifted CG ライブラリの動作イメージ

COCG と同様.

Chapter 5

使用方法

各ライブラリともユーザーはライブラリ名および型を指定し、

- 初期設定 (init 関数)
- アップデート (update 関数)
- 計算情報出力 (getcoef, getvec 関数などを呼び出力)
- 終了関数 (finalize 関数)

の手順で関数を使用することで、計算が実施される。なお、リスタートを行う場合には

- 初期設定関数 (restart 関数)
- アップデート (update 関数)
- 計算情報出力 (getcoef, getvec 関数などを呼び出力)
- 終了関数 (finalize 関数)

の手順で実行する。fortran から呼び出すときには

```
USE komega_????
```

のようにモジュールを呼び出す。"????" の部分には, "CG_R", "CG_C", "COCG", "BiCG" が入る。MPI/Hybrid 並列版のルーチンを利用するときには、

```
USE pkomega_????
```

のようになる。

C/C++で書かれたプログラムから呼び出すときには、

```
#include komega_????.h
```

のようにヘッダーファイルを読み込む。また、スカラー引数はすべてポインタとして渡す。MPI/Hybrid 並列版のルーチンを利用するときには、

```
#include pkomega_????.h
```

のようになる。またライブラリに渡すコミュニケーター変数を、次のように C/C++のものから fortran のものに変換する。

```
comm_f = MPI_Comm_c2f(comm_c);
```


5.1 各ルーチンの説明

5.1.1 komega_????_init, pkomega_????_init

ライブラリ内部変数の割り付けおよび初期化を行う。シフト線形方程式を解く前に、一番初めに実行する。

構文

Fortran シリアル/OpenMP 版

```
CALL komega_CG_R_init(ndim, nl, nz, x, z, itermax, threshold)
CALL komega_CG_C_init(ndim, nl, nz, x, z, itermax, threshold)
CALL komega_COCG_init(ndim, nl, nz, x, z, itermax, threshold)
CALL komega_BiCG_init(ndim, nl, nz, x, z, itermax, threshold)
```

Fortran MPI/Hybrid 並列版

```
CALL pkomega_CG_R_init(ndim, nl, nz, x, z, itermax, threshold, comm)
CALL pkomega_CG_C_init(ndim, nl, nz, x, z, itermax, threshold, comm)
CALL pkomega_COCG_init(ndim, nl, nz, x, z, itermax, threshold, comm)
CALL pkomega_BiCG_init(ndim, nl, nz, x, z, itermax, threshold, comm)
```

C/C++ シリアル/OpenMP 版

```
komega_CG_R_init(&ndim, &nl, &nz, x, z, &itermax, &threshold);
komega_CG_C_init(&ndim, &nl, &nz, x, z, &itermax, &threshold);
komega_COCG_init(&ndim, &nl, &nz, x, z, &itermax, &threshold);
komega_BiCG_init(&ndim, &nl, &nz, x, z, &itermax, &threshold);
```

C/C++ MPI/Hybrid 並列版

```
pkomega_CG_R_init(&ndim, &nl, &nz, x, z, &itermax, &threshold, &comm);
pkomega_CG_C_init(&ndim, &nl, &nz, x, z, &itermax, &threshold, &comm);
pkomega_COCG_init(&ndim, &nl, &nz, x, z, &itermax, &threshold, &comm);
pkomega_BiCG_init(&ndim, &nl, &nz, x, z, &itermax, &threshold, &comm);
```

パラメーター

- ndim
INTEGER. スカラー. 入力. 線形方程式の次元.
- nl
INTEGER. スカラー. 入力. 射影された解ベクトルの次元.
- nz
INTEGER. スカラー. 入力. シフト点の数.
- x
DOUBLE PRECISION (CG_R_init の場合), DOUBLE COMPLEX (それ以外). 長さ nl*nz の配列. 出力. 解ベクトル. 0 ベクトルが返される.
- z
DOUBLE PRECISION (CG_R_init, CG_C_init の場合), DOUBLE COMPLEX (それ以外). 長さ nz の配列. 入力. シフト点.
- itermax
INTEGER. スカラー. 入力. リスタート用配列の割り付けのための最大反復回数. これを 0 にした場合にはリスタート用配列を割りつけない (したがって後述のリスタート用変数の出力を行えない)

- threshold

DOUBLE PRECISION. スカラー. 入力. 収束判定用しきい値. シード方程式の残差ベクトルの 2-ノルムがこの値を下回った時に収束したと判定する.

- comm

INTEGER. スカラー. 入力. MPI/Hybrid 並列版のみ. MPI のコミュニケーター (MPI_COMM_WORLD など) を入れる.

5.1.2 komega_????_restart, pkomega_????_restart

リスタートを行う場合に ?_init の代わりに用いる. ライブラリ内部変数の割り付けおよび初期化を行う. シフト線形方程式を解く前に, 一番初めに実行する.

構文

Fortran (シリアル/OpenMP 版)

```
CALL komega_CG_R_restart(ndim, nl, nz, x, z, itermax, threshold, status, &
& iter_old, v2, v12, alpha_save, beta_save, z_seed, r_l_save)
CALL komega_CG_C_restart(ndim, nl, nz, x, z, itermax, threshold, status, &
& iter_old, v2, v12, alpha_save, beta_save, z_seed, r_l_save)
CALL komega_COCG_restart(ndim, nl, nz, x, z, itermax, threshold, status, &
& iter_old, v2, v12, alpha_save, beta_save, z_seed, r_l_save)
CALL komega_BiCG_restart(ndim, nl, nz, x, z, itermax, threshold, status, &
& iter_old, v2, v12, v4, v14, alpha_save, beta_save, &
& z_seed, r_l_save)
```

Fortran (MPI/ハイブリッド並列版)

```
CALL pkomega_CG_R_restart(ndim, nl, nz, x, z, itermax, threshold, comm, status, &
& iter_old, v2, v12, alpha_save, beta_save, z_seed, r_l_save)
CALL pkomega_CG_C_restart(ndim, nl, nz, x, z, itermax, threshold, comm, status, &
& iter_old, v2, v12, alpha_save, beta_save, z_seed, r_l_save)
CALL pkomega_COCG_restart(ndim, nl, nz, x, z, itermax, threshold, comm, status, &
& iter_old, v2, v12, alpha_save, beta_save, z_seed, r_l_save)
CALL pkomega_BiCG_restart(ndim, nl, nz, x, z, itermax, threshold, comm, status, &
& iter_old, v2, v12, v4, v14, alpha_save, beta_save, &
& z_seed, r_l_save)
```

C/C++ (シリアル/OpenMP 版)

```
komega_CG_R_restart(&ndim, &nl, &nz, x, z, &itermax, &threshold, status, &
& &iter_old, v2, v12, alpha_save, beta_save, &z_seed, r_l_save);
komega_CG_C_restart(&ndim, &nl, &nz, x, z, &itermax, &threshold, status, &
& &iter_old, v2, v12, alpha_save, beta_save, &z_seed, r_l_save);
komega_COCG_restart(&ndim, &nl, &nz, x, z, &itermax, &threshold, status, &
& &iter_old, v2, v12, alpha_save, beta_save, &z_seed, r_l_save);
komega_BiCG_restart(&ndim, &nl, &nz, x, z, &itermax, &threshold, status, &
& &iter_old, v2, v12, v4, v14, alpha_save, beta_save, &
& &z_seed, r_l_save);
```

C/C++ (MPI/ハイブリッド並列版)

```
pkomega_CG_R_restart(&ndim, &nl, &nz, x, z, &itermax, &threshold, &comm, status, &
& &iter_old, v2, v12, alpha_save, beta_save, &z_seed, r_l_save);
pkomega_CG_C_restart(&ndim, &nl, &nz, x, z, &itermax, &threshold, &comm, status, &
& &iter_old, v2, v12, alpha_save, beta_save, &z_seed, r_l_save);
pkomega_COCG_restart(&ndim, &nl, &nz, x, z, &itermax, &threshold, &comm, status, &
```

```

&                &iter_old, v2, v12, alpha_save, beta_save, &z_seed, r_l_save);
pkomega_BiCG_restart(&ndim, &nl, &nz, x, z, &itermax, &threshold, &comm, status, &
&                &iter_old, v2, v12, v4, v14, alpha_save, beta_save, &
&                &z_seed, r_l_save);

```

パラメーター

- `ndim, nl, nz, x, z, itermax, threshold, comm`

?_init と同様.

- `status`

INTEGER. 長さ 3 の配列. 出力. エラーコードを返す.

第一成分 (**status(1)**) 解が収束した場合, もしくは計算が破綻した場合には現在の総反復回数にマイナスが付いた値が返される. それ以外の場合には現在の総反復回数 (マイナスが付かない) が返される. `status(1)` が正の値の時のみ反復を続行できる. それ以外の場合は反復を進めても有意な結果は得られない.

第二成分 (**status(2)**) `itermax` を有限にして, かつ `itermax` 回の反復で収束に達しなかった場合には 1 が返される. α が発散した場合には 2 が返される. π_{seed} が 0 になった場合には 3 が返される. `COCG_restart` もしくは `BiCG_restart` で, 残差ベクトルと影の残差ベクトルが直交した場合には 4 が返される. それ以外の場合には 0 が返される.

第三成分 (**status(3)**) シード点の `index` が返される.

- `iter_old`

INTEGER. スカラー. 入力. 先行する計算での反復回数.

- `v2`

DOUBLE PRECISION (CG_R_restart の場合), DOUBLE COMPLEX (それ以外). 長さ `ndim` の配列. 入力. 先行する計算での最後の残差ベクトル.

- `v12`

DOUBLE PRECISION (CG_R_restart の場合), DOUBLE COMPLEX (それ以外). 長さ `ndim` の配列. 入力. 先行する計算での最後から 2 番目の残差ベクトル.

- `alpha_save`

DOUBLE PRECISION (CG_R_restart, CG_C_restart の場合), DOUBLE COMPLEX (それ以外). 長さ `iter_old` の配列. 入力. 先行する計算での各反復での (Bi)CG 法のパラメーター α .

- `beta_save`

DOUBLE PRECISION (CG_R_restart, CG_C_restart の場合), DOUBLE COMPLEX (それ以外). 長さ `iter_old` の配列. 入力. 先行する計算での各反復での (Bi)CG 法のパラメーター β .

- `z_seed`

DOUBLE PRECISION (CG_R_restart, CG_C_restart の場合), DOUBLE COMPLEX (それ以外). スカラー. 入力. 先行する計算でのシードシフト.

- `r_l_save`

DOUBLE PRECISION (CG_R_restart の場合), DOUBLE COMPLEX (それ以外). 長さ `nl*iter_old` の配列. 入力. 先行する計算での各反復での射影された残差ベクトル.

- `v4`

BiCG_restart の場合のみ使用. DOUBLE COMPLEX. 長さ `ndim` の配列. 入力. 先行する計算での最後の影の残差ベクトル.

- v14

BiCG_restart の場合のみ使用. DOUBLE COMPLEX. 長さ ndim の配列. 入力. 先行する計算での最後から 2 番目の影の残差ベクトル.

5.1.3 komega_????_update, pkomega_????_update

ループ内で行列ベクトル積と交互に呼ばれて解を更新する.

構文

Fortran (シリアル/OpenMPI 版)

```
CALL komega_CG_R_update(v12, v2, x, r_l, status)
CALL komega_CG_C_update(v12, v2, x, r_l, status)
CALL komega_COCG_update(v12, v2, x, r_l, status)
CALL komega_BiCG_update(v12, v2, v14, v4, x, r_l, status)
```

Fortran (MPI/ハイブリッド並列版)

```
CALL pkomega_CG_R_update(v12, v2, x, r_l, status)
CALL pkomega_CG_C_update(v12, v2, x, r_l, status)
CALL pkomega_COCG_update(v12, v2, x, r_l, status)
CALL pkomega_BiCG_update(v12, v2, v14, v4, x, r_l, status)
```

C/C++ (シリアル/OpenMPI 版)

```
komega_CG_R_update(v12, v2, x, r_l, status);
komega_CG_C_update(v12, v2, x, r_l, status);
komega_COCG_update(v12, v2, x, r_l, status);
komega_BiCG_update(v12, v2, v14, v4, x, r_l, status);
```

C/C++ (MPI/ハイブリッド並列版)

```
pkomega_CG_R_update(v12, v2, x, r_l, status);
pkomega_CG_C_update(v12, v2, x, r_l, status);
pkomega_COCG_update(v12, v2, x, r_l, status);
pkomega_BiCG_update(v12, v2, v14, v4, x, r_l, status);
```

パラメーター

- v12

DOUBLE PRECISION (CG_R_update の場合), DOUBLE COMPLEX (それ以外). 長さ ndim の配列. 入出力. 入力: 残差ベクトル (v2) と行列の積. 出力は, 更新された残差ベクトルの 2-ノルムが, 先頭の要素に格納される (これは収束の具合を表示して調べる時に用いる).

- v2

DOUBLE PRECISION (CG_R_update の場合), DOUBLE COMPLEX (それ以外). 長さ ndim の配列. 入出力. 入力: 残差ベクトル. 出力は更新された残差ベクトル.

- v14

BiCG_update の場合のみ使用. DOUBLE COMPLEX. 長さ ndim の配列. 入力. 影の残差ベクトル (v4) と行列の積.

- v4

BiCG_update の場合のみ使用. DOUBLE COMPLEX. 長さ ndim の配列. 入出力. 入力: 影の残差ベクトル. 出力は更新された影の残差ベクトル.

- status

INTEGER. 長さ 3 の配列. 出力. エラーコードを返す.

第一成分 (**status(1)**) 解が収束した場合, もしくは計算が破綻した場合には現在の総反復回数にマイナスが付いた値が返される. それ以外の場合には現在の総反復回数 (マイナスが付かない) が返される. **status(1)** が正の値の時のみ反復を続行できる. それ以外の場合は反復を進めても有意な結果は得られない.

第二成分 (**status(2)**) `?_init` ルーチンで, `itermax` を有限にして, かつ `itermax` 回の反復で収束に達しなかった場合には 1 が返される. α が発散した場合には 2 が返される. π_{seed} が 0 になった場合には 3 が返される. `COCG_update` もしくは `BiCG_update` で, 残差ベクトルと影の残差ベクトルが直交した場合には 4 が返される. それ以外の場合には 0 が返される.

第三成分 (**status(3)**) シード点の index が返される.

5.1.4 komega_????_getcoef, pkomega_????_getcoef

後でリスタートをするときに必要な係数を取得する. このルーチン呼び出すためには, `?_init` ルーチンで `itermax` を 0 以外の値にしておく必要がある.

構文

Fortran (シリアル/OpenMP 版)

```
CALL komega_CG_R_getcoef(alpha_save, beta_save, z_seed, r_l_save)
CALL komega_CG_C_getcoef(alpha_save, beta_save, z_seed, r_l_save)
CALL komega_COCG_getcoef(alpha_save, beta_save, z_seed, r_l_save)
CALL komega_BiCG_getcoef(alpha_save, beta_save, z_seed, r_l_save)
```

Fortran (MPI/ハイブリッド並列版)

```
CALL pkomega_CG_R_getcoef(alpha_save, beta_save, z_seed, r_l_save)
CALL pkomega_CG_C_getcoef(alpha_save, beta_save, z_seed, r_l_save)
CALL pkomega_COCG_getcoef(alpha_save, beta_save, z_seed, r_l_save)
CALL pkomega_BiCG_getcoef(alpha_save, beta_save, z_seed, r_l_save)
```

C/C++ (シリアル/OpenMP 版)

```
komega_CG_R_getcoef(alpha_save, beta_save, &z_seed, r_l_save);
komega_CG_C_getcoef(alpha_save, beta_save, &z_seed, r_l_save);
komega_COCG_getcoef(alpha_save, beta_save, &z_seed, r_l_save);
komega_BiCG_getcoef(alpha_save, beta_save, &z_seed, r_l_save);
```

C/C++ (MPI/ハイブリッド並列版)

```
pkomega_CG_R_getcoef(alpha_save, beta_save, &z_seed, r_l_save);
pkomega_CG_C_getcoef(alpha_save, beta_save, &z_seed, r_l_save);
pkomega_COCG_getcoef(alpha_save, beta_save, &z_seed, r_l_save);
pkomega_BiCG_getcoef(alpha_save, beta_save, &z_seed, r_l_save);
```

パラメーター

- alpha_save

DOUBLE PRECISION (CG_R_getcoef, CG_C_getcoef の場合), DOUBLE COMPLEX (それ以外). 総反復回数と同じ長さの配列. 出力. 各反復での (Bi)CG 法のパラメーター α .

- beta_save

DOUBLE PRECISION (CG_R_getcoef, CG_C_getcoef の場合), DOUBLE COMPLEX (それ以外). 総反復回数と同じ長さの配列. 出力. 各反復での (Bi)CG 法のパラメーター β .

- `z_seed`

DOUBLE PRECISION (CG_R_getoef, CG_C_getoef の場合), DOUBLE COMPLEX (それ以外). スカラー. 出力. シードシフト.

- `r_l_save`

DOUBLE PRECISION (CG_R_getoef の場合), DOUBLE COMPLEX (それ以外). $n_l \times$ 総反復回数の長さ配列. 出力. 各反復での射影された残差ベクトル.

5.1.5 `komega_????_getvec`, `pkomega_????_getvec`

後でリスタートをするときに必要な残差ベクトルを取得する. このルーチンを呼び出すためには, `?_init` ルーチンで `itermax` を 0 以外の値にしておく必要がある.

構文

Fortran (シリアル/OpenMP 版)

```
CALL komega_CG_R_getvec(r_old)
CALL komega_CG_C_getvec(r_old)
CALL komega_COCG_getvec(r_old)
CALL komega_BiCG_getvec(r_old, r_tilde_old)
```

Fortran (MPI/ハイブリッド並列版)

```
CALL pkomega_CG_R_getvec(r_old)
CALL pkomega_CG_C_getvec(r_old)
CALL pkomega_COCG_getvec(r_old)
CALL pkomega_BiCG_getvec(r_old, r_tilde_old)
```

C/C++ (シリアル/OpenMP 版)

```
komega_CG_R_getvec(r_old);
komega_CG_C_getvec(r_old);
komega_COCG_getvec(r_old);
komega_BiCG_getvec(r_old, r_tilde_old);
```

C/C++ (MPI/ハイブリッド並列版)

```
pkomega_CG_R_getvec(r_old);
pkomega_CG_C_getvec(r_old);
pkomega_COCG_getvec(r_old);
pkomega_BiCG_getvec(r_old, r_tilde_old);
```

パラメーター

- `r_old`

DOUBLE PRECISION (CG_R_getvec の場合), DOUBLE COMPLEX (それ以外). 長さ `ndim` の配列. 出力. 先行する計算での最後から 2 番目の残差ベクトル.

- `r_tilde_old`

BiCG_getvec の場合のみ使用. DOUBLE COMPLEX. 長さ `ndim` の配列. 出力. 先行する計算での最後から 2 番目の影の残差ベクトル.

5.1.6 komega_????_getresidual, pkomega_????_getresidual

各シフト点での残差ベクトルの 2-ノルムを取得する。このルーチンは komega_????_init と komega_????_finalize の間の任意の場所で呼び出すことができる。また、いつ何回呼び出しても最終的な計算結果には影響を与えない。

構文

Fortran (シリアル/OpenMP 版)

```
CALL komega_CG_R_getresidual(res)
CALL komega_CG_C_getresidual(res)
CALL komega_COEG_getresidual(res)
CALL komega_BiCG_getresidual(res)
```

Fortran (MPI/ハイブリッド並列版)

```
CALL pkomega_CG_R_getresidual(res)
CALL pkomega_CG_C_getresidual(res)
CALL pkomega_COEG_getresidual(res)
CALL pkomega_BiCG_getresidual(res)
```

C/C++ (シリアル/OpenMP 版)

```
komega_CG_R_getresidual(res);
komega_CG_C_getresidual(res);
komega_COEG_getresidual(res);
komega_BiCG_getresidual(res);
```

C/C++ (MPI/ハイブリッド並列版)

```
pkomega_CG_R_getresidual(res);
pkomega_CG_C_getresidual(res);
pkomega_COEG_getresidual(res);
pkomega_BiCG_getresidual(res);
```

パラメーター

- res

DOUBLE PRECISION. 長さ nz の配列. 出力. 各シフト点での残差ベクトルの 2-ノルム.

5.1.7 komega_????_finalize, pkomega_????_finalize

ライブラリ内部で割りつけた配列のメモリを解放する。

構文

Fortran (シリアル/OpenMP 版)

```
CALL komega_CG_R_finalize()
CALL komega_CG_C_finalize()
CALL komega_COEG_finalize()
CALL komega_BiCG_finalize()
```

Fortran (MPI/ハイブリッド並列版)

```
CALL pkomega_CG_R_finalize()
CALL pkomega_CG_C_finalize()
CALL pkomega_COEG_finalize()
CALL pkomega_BiCG_finalize()
```

C/C++ (シリアル/OpenMP 版)

```
komega_CG_R_finalize();  
komega_CG_C_finalize();  
komega_COCG_finalize();  
komega_BiCG_finalize();
```

C/C++ (MPI/ハイブリッド並列版)

```
pkomega_CG_R_finalize();  
pkomega_CG_C_finalize();  
pkomega_COCG_finalize();  
pkomega_BiCG_finalize();
```

5.2 Shifted BiCG ライブラリを使用したソースコードの例

以下, 代表的な例として Shifted BiCG ライブラリの場合の使用方法を記載する.

PROGRAM my_prog

```
!  
USE komega_bicg, ONLY : komega_BiCG_init, komega_BiCG_restart, &  
& komega_BiCG_update, komega_BiCG_getcoef, &  
& komega_BiCG_getvec, komega_BiCG_finalize  
USE solve_cc_routines, ONLY : input_size, input_restart, &  
& projection, &  
& hamiltonian_prod, generate_system, &  
& output_restart, output_result  
!  
IMPLICIT NONE  
!  
INTEGER, SAVE :: &  
& rnd_seed, &  
& ndim, & ! Size of Hilvert space  
& nz, & ! Number of frequencies  
& nl, & ! Number of Left vector  
& itermax, & ! Max. number of iteration  
& iter_old ! Number of iteration of previous run  
!  
REAL(8), SAVE :: &  
& threshold ! Convergence Threshold  
!  
COMPLEX(8), SAVE :: &  
& z_seed ! Seed frequency  
!  
COMPLEX(8), ALLOCATABLE, SAVE :: &  
& z(:) ! (nz): Frequency  
!  
COMPLEX(8), ALLOCATABLE, SAVE :: &  
& ham(:, :), &  
& rhs(:), &  
& v12(:), v2(:), & ! (ndim): Working vector  
& v14(:), v4(:), & ! (ndim): Working vector  
& r_1(:), & ! (nl) : Projected residual vector  
& x(:, :) ! (nl, nz) : Projected result  
!  
! Variables for Restart  
!
```



```

COMPLEX(8),ALLOCATABLE,SAVE :: &
& alpha(:), beta(:) ! (iter_old)
!
COMPLEX(8),ALLOCATABLE,SAVE :: &
& r_l_save(:, :) ! (nl, iter_old) Projected residual vectors
!
! Variables for Restart
!
INTEGER :: &
& itermin, & ! First iteration in this run
& iter, & ! Counter for Iteration
& status(3)
!
! Input Size of vectors
!
CALL input_size(ndim, nl, nz)
!
ALLOCATE(v12(ndim), v2(ndim), v14(ndim), v4(ndim), r_l(nl), &
& x(nl, nz), z(nz), ham(ndim, ndim), rhs(ndim))
!
CALL generate_system(ndim, ham, rhs, z)
!
! Check: Whether the restart file is exist.
!
CALL input_restart(iter_old, zseed, alpha, beta, r_l_save)
!
WRITE(*, *)
WRITE(*, *) "##### CG Initialization #####"
WRITE(*, *)
!
IF(iter_old > 0) THEN
!
! When restarting, counter
!
itermin = iter_old + 1
CALL komega_BiCG_restart(ndim, nl, nz, x, z, max(0, itermax), &
& threshold, &
& status, iter_old, v2, v12, v4, v14, alpha, &
& beta, z_seed, r_l_save)
!
! These vectors were saved in BiCG routine
!
DEALLOCATE(alpha, beta, r_l_save)
!
IF(status(1) /= 0) GOTO 10
!
ELSE
!
itermin = 1
!
! Generate Right Hand Side Vector
!
v2(1:ndim) = rhs(1:ndim)
v4(1:ndim) = CONJG(v2(1:ndim))
!v4(1:ndim) = v2(1:ndim)
!
CALL komega_BiCG_init(ndim, nl, nz, x, z, max(0, itermax), &
& threshold)

```

```
!
END IF
!
! BiCG Loop
!
WRITE(*,*)
WRITE(*,*) "##### CG Iteration  #####"
WRITE(*,*)
!
DO iter = 1, abs(itermax)
!
! Projection of Residual vector into the space
! spaned by left vectors
!
r_l(1:nl) = projection(v2(1:nl))
!
! Matrix-vector product
!
CALL hamiltonian_prod(Ham, v2, v12)
CALL hamiltonian_prod(Ham, v4, v14)
!
! Update result x with BiCG
!
CALL komega_BiCG_update(v12, v2, v14, v4, x, r_l, status)
!
WRITE(*,'(a,i,a,3i,a,e15.5)') "lopp : ", iter, &
&                                ", status : ", status(1:3), &
&                                ", Res. : ", DBLE(v12(1))
IF(status(1) < 0) EXIT
!
END DO
!
IF(status(2) == 0) THEN
WRITE(*,*) " Converged in iteration ", ABS(status(1))
ELSE IF(status(2) == 1) THEN
WRITE(*,*) " Not Converged in iteration ", ABS(status(1))
ELSE IF(status(2) == 2) THEN
WRITE(*,*) " Alpha becomes infinity", ABS(status(1))
ELSE IF(status(2) == 3) THEN
WRITE(*,*) " Pi_seed becomes zero", ABS(status(1))
ELSE IF(status(2) == 4) THEN
WRITE(*,*) " Residual & Shadow residual are orthogonal", &
&          ABS(status(1))
END IF
iter_old = ABS(status(1))
!
! Get these vectors for restart in the Next run
!
IF(itermax > 0) THEN
!
ALLOCATE(alpha(iter_old), beta(iter_old), r_l_save(nl,iter_old))
!
CALL komega_BiCG_getcoef(alpha, beta, z_seed, r_l_save)
CALL komega_BiCG_getvec(v12,v14)
!
CALL output_restart(iter_old, z_seed, alpha, beta, &
&                    r_l_save, v12, v14)
!
```

```
        DEALLOCATE(alpha, beta, r_l_save)
        !
    END IF
    !
10 CONTINUE
    !
    ! Deallocate all intrinsic vectors
    !
    CALL komega_BiCG_finalize()
    !
    ! Output to a file
    !
    CALL output_result(nl, nz, z, x, r_l)
    !
    DEALLOCATE(vl2, v2, vl4, v4, r_l, x, z)
    !
    WRITE(*,*)
    WRITE(*,*) "##### Done #####"
    WRITE(*,*)
    !
END PROGRAM my_prog
```

Chapter 6

Contact

このライブラリについてのご意見, ご質問, バグ報告等ありましたら下記までお問い合わせください。

河村光晶

`mkawamura_at_issp.u-tokyo.ac.jp`

`_at_`を`@`に変えてください.

Bibliography

[Flommer2003] 1. Frommer, Computing **70**, 87 (2003).

[Yamamoto2008] 19. Yamamoto, T. Sogabe, T. Hoshi, S.-L. Zhang, and T. Fujiwara, J. Phys. Soc. Jpn. **77**, 114713 (2008).